NAG Fortran Library Routine Document

F08BAF (DGELSY)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F08BAF (DGELSY) computes the minimum-norm solution to a real linear least-squares problem

$$\min_x \|b - Ax\|_2$$

using a complete orthogonal factorization of A. A is an m by n matrix which may be rank-deficient. Several right-hand side vectors b and solution vectors x can be handled in a single call.

2 Specification

```
SUBROUTINE F08BAF (M, N, NRHS, A, LDA, B, LDB, JPVT, RCOND, RANK, WORK,
1                  LWORK, INFO)
INTEGER            M, N, NRHS, LDA, LDB, JPVT(*), RANK, LWORK, INFO
double precision   A(LDA,*), B(LDB,*), RCOND, WORK(*)
```

The routine may be called by its LAPACK name ***dgelsy***.

3 Description

The right-hand side vectors are stored as the columns of the m by r matrix B and the solution vectors in the n by r matrix X.

F08BAF (DGELSY) first computes a QR factorization with column pivoting

$$AP = Q\begin{pmatrix} R_{11} & R_{12} \\ 0 & R_{22} \end{pmatrix},$$

with R_{11} defined as the largest leading submatrix whose estimated condition number is less than $1/\mathrm{RCOND}$. The order of R_{11}, RANK, is the effective rank of A.

Then, R_{22} is considered to be negligible, and R_{12} is annihilated by orthogonal transformations from the right, arriving at the complete orthogonal factorization

$$AP = Q\begin{pmatrix} T_{11} & 0 \\ 0 & 0 \end{pmatrix}Z.$$

The minimum-norm solution is then

$$X = PZ^T\begin{pmatrix} T_{11}^{-1}{Q_1}^T b \\ 0 \end{pmatrix}$$

where Q_1 consists of the first RANK columns of Q.

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia URL: http://www.netlib.org/lapack/lug

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Parameters

1: M – INTEGER *Input*

On entry: m, the number of rows of the matrix A.

Constraint: $\mathrm{M} \geq 0$.

2: N – INTEGER *Input*

On entry: n, the number of columns of the matrix A.

Constraint: $\mathrm{N} \geq 0$.

3: NRHS – INTEGER *Input*

On entry: r, the number of right-hand sides, i.e., the number of columns of the matrices B and X.

Constraint: $\mathrm{NRHS} \geq 0$.

4: A(LDA,$*$) – ***double precision*** array *Input/Output*

Note: the second dimension of the array A must be at least $\max(1, \mathrm{N})$.

On entry: the m by n matrix A.

On exit: has been overwritten by details of its complete orthogonal factorisation.

5: LDA – INTEGER *Input*

On entry: the first dimension of the array A as declared in the (sub)program from which F08BAF (DGELSY) is called.

Constraint: $\mathrm{LDA} \geq \max(1, \mathrm{M})$.

6: B(LDB,$*$) – ***double precision*** array *Input/Output*

Note: the second dimension of the array B must be at least $\max(1, \mathrm{NRHS})$.

On entry: the m by r right-hand side matrix B.

On exit: the n by r solution matrix X.

7: LDB – INTEGER *Input*

On entry: the first dimension of the array B as declared in the (sub)program from which F08BAF (DGELSY) is called.

Constraint: $\mathrm{LDB} \geq \max(1, \mathrm{M}, \mathrm{N})$.

8: JPVT($*$) – INTEGER array *Input/Output*

Note: the dimension of the array JPVT must be at least $\max(1, \mathrm{N})$.

On entry: if $\mathrm{JPVT}(i) \neq 0$, the ith column of A is permuted to the front of AP, otherwise column i is a free column.

On exit: if $\mathrm{JPVT}(i) = k$, then the ith column of AP was the kth column of A.

9: RCOND – ***double precision*** *Input*

On entry: used to determine the effective rank of A, which is defined as the order of the largest leading triangular submatrix R_{11} in the QR factorization of A, whose estimated condition number is $< 1/\mathrm{RCOND}$.

10: RANK – INTEGER *Output*

On exit: the effective rank of A, i.e., the order of the submatrix R_{11}. This is the same as the order of the submatrix T_{11} in the complete orthogonal factorization of A.

11: WORK$(*)$ – ***double precision*** array *Workspace*

Note: the dimension of the array WORK must be at least $\max(1, \mathrm{LWORK})$.

On exit: if $\mathrm{INFO} = 0$, $\mathrm{WORK}(1)$ returns the optimal LWORK.

12: LWORK – INTEGER *Input*

On entry: the dimension of the array WORK as declared in the (sub)program from which F08BAF (DGELSY) is called.

For optimal performance,

$$\mathrm{LWORK} \geq \max(k + 2 \times \mathrm{N} + nb \times (\mathrm{N} + 1), 2 \times k + nb \times \mathrm{NRHS}),$$

where $k = \min(\mathrm{M}, \mathrm{N})$ and nb is the optimal block size.

If $\mathrm{LWORK} = -1$, a workspace query is assumed; the routine only calculates the optimal size of the WORK array, returns this value as the first entry of the WORK array, and no error message related to LWORK is issued.

Constraint: $\mathrm{LWORK} \geq \max(k + 3 \times \mathrm{N} + 1, 2 \times k + \mathrm{NRHS})$, where $k = \min(\mathrm{M}, \mathrm{N})$.

13: INFO – INTEGER *Output*

On exit: $\mathrm{INFO} = 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

$\mathrm{INFO} < 0$

If $\mathrm{INFO} = -i$, the ith argument had an illegal value.

7 Accuracy

See Section 4.5 of Anderson *et al.* (1999) for details of error bounds.

8 Further Comments

The complex analogue of this routine is F08BNF (ZGELSY).

9 Example

To solve the linear least squares problem

$$\min_x \|b - Ax\|_2$$

for the solution, x, of minimum norm, where

$$A = \begin{pmatrix} -0.09 & 0.14 & -0.46 & 0.68 & 1.29 \\ -1.56 & 0.20 & 0.29 & 1.09 & 0.51 \\ -1.48 & -0.43 & 0.89 & -0.71 & -0.96 \\ -1.09 & 0.84 & 0.77 & 2.11 & -1.27 \\ 0.08 & 0.55 & -1.13 & 0.14 & 1.74 \\ -1.59 & -0.72 & 1.06 & 1.24 & 0.34 \end{pmatrix}$$

and

$$b = \begin{pmatrix} 7.4 \\ 4.2 \\ -8.3 \\ 1.8 \\ 8.6 \\ 2.1 \end{pmatrix}.$$

A tolerance of 0.01 is used to determine the effective rank of A.

Note that the block size (NB) of 64 assumed in this example is not realistic for such a small problem, but should be suitable for large problems.

9.1 Program Text

Note: the listing of the example program presented below uses ***bold italicised*** terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     F08BAF Example Program Text
*     Mark 21 Release. NAG Copyright 2004.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          MMAX, NB, NMAX
      PARAMETER        (MMAX=16,NB=64,NMAX=8)
      INTEGER          LDA, LWORK
      PARAMETER        (LDA=MMAX,LWORK=3*NMAX+NB*(NMAX+1))
*     .. Local Scalars ..
      DOUBLE PRECISION RCOND
      INTEGER          I, INFO, J, M, N, RANK
*     .. Local Arrays ..
      DOUBLE PRECISION A(LDA,NMAX), B(MMAX), WORK(LWORK)
      INTEGER          JPVT(NMAX)
*     .. External Subroutines ..
      EXTERNAL         DGELSY, F06DBF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F08BAF Example Program Results'
      WRITE (NOUT,*)
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) M, N
      IF (M.LE.MMAX .AND. N.LE.NMAX .AND. M.GE.N) THEN
*
*        Read A and B from data file
*
         READ (NIN,*) ((A(I,J),J=1,N),I=1,M)
         READ (NIN,*) (B(I),I=1,M)
*
*        Initialize JPVT to be zero so that all columns are free
*
         CALL F06DBF(N,0,JPVT,1)
*
*        Choose RCOND to reflect the relative accuracy of the input data
*
         RCOND = 0.01D0
*
*        Solve the least squares problem min( norm2(b - Ax) ) for the x
*        of minimum norm.
*
         CALL DGELSY(M,N,1,A,LDA,B,M,JPVT,RCOND,RANK,WORK,LWORK,INFO)
*
*        Print solution
*
         WRITE (NOUT,*) 'Least squares solution'
         WRITE (NOUT,99999) (B(I),I=1,N)
*
*        Print the effective rank of A
*
```

```
         WRITE (NOUT,*)
         WRITE (NOUT,*) 'Tolerance used to estimate the rank of A'
         WRITE (NOUT,99998) RCOND
         WRITE (NOUT,*) 'Estimated rank of A'
         WRITE (NOUT,99997) RANK
      ELSE
         WRITE (NOUT,*) 'MMAX and/or NMAX too small, and/or M.LT.N'
      END IF
      STOP
*
99999 FORMAT (1X,7F11.4)
99998 FORMAT (3X,1P,E11.2)
99997 FORMAT (1X,I6)
      END
```

9.2 Program Data

```
F08BAF Example Program Data

  6      5                          :Values of M and N

 -0.09   0.14  -0.46   0.68   1.29
 -1.56   0.20   0.29   1.09   0.51
 -1.48  -0.43   0.89  -0.71  -0.96
 -1.09   0.84   0.77   2.11  -1.27
  0.08   0.55  -1.13   0.14   1.74
 -1.59  -0.72   1.06   1.24   0.34 :End of matrix A

  7.4
  4.2
 -8.3
  1.8
  8.6
  2.1                               :End of vector b
```

9.3 Program Results

```
 F08BAF Example Program Results

 Least squares solution
      0.6344     0.9699    -1.4402     3.3678     3.3992

 Tolerance used to estimate the rank of A
      1.00E-02
 Estimated rank of A
      4
```